Temperatura LM35+Arduino
An analog temperature sensor is pretty easy to explain, its a chip that tells you what the ambient temperature is!



These sensors use a solid-state technique to determine the temperature. That is to say, they don't use mercury (like old thermometers), bimetalic strips (like in some home thermometers or stoves), nor do they use thermistors (temperature sensitive resistors). Instead, they use the fact as temperature increases, the voltage across a diode increases at a known rate. (Technically, this is actually the voltage drop between the base and emitter - the Vbe - of a transistor.) By precisely amplifying the voltage change, it is easy to generate an analog signal that is directly proportional to temperature. There have been some improvements on the technique but, essentially that is how temperature is measured.



Because these sensors have no moving parts, they are precise, never wear out, don't need calibration, work under many environmental conditions, and are consistant between sensors and readings. Moreover they are very inexpensive and quite easy to use.

Some Basic Stats

These stats are for the temperature sensor in the Adafruit shop, the Analog Devices TMP36 (-40 to 150C). Its very similar to the LM35/TMP35 (Celsius output) and LM34/TMP34 (Farenheit output). The reason we went with the '36 instead of the '35 or '34 is that this sensor has a very wide range and doesn't require a negative voltage to read sub-zero temperatures. Otherwise, the functionality is basically the same.

· Size: TO-92 package (about 0.2" x 0.2" x 0.2") with three leads

· Price: $2.00 at the Adafruit shop
· Temperature range: -40°C to 150°C / -40°F to 302°F

· Output range: 0.1V (-40°C) to 2.0V (150°C) but accuracy decreases after 125°C

· Power supply: 2.7V to 5.5V only, 0.05 mA current draw

· Datasheet
How to Measure Temperature

Using the TMP36 is easy, simply connect the left pin to power (2.7-5.5V) and the right pin to ground. Then the middle pin will have an analog voltage that is directly proportional (linear) to the temperature. The analog voltage is independant of the power supply.



To convert the voltage to temperature, simply use the basic formula:

Temp in °C = [(Vout in mV) - 500] / 10
So for example, if the voltage out is 1V that means that the temperature is ((1000 mV - 500) / 10) = 50 °C
If you're using a LM35 or similar, use line 'a' in the image above and the formula: Temp in °C = (Vout in mV) / 10
Problems you may encounter with multiple sensors:

If, when adding more sensors, you find that the temperature is inconsistant, this indicates that the sensors are interfering with each other when switching the analog reading circuit from one pin to the other. You can fix this by doing two delayed readings and tossing out the first one.

Testing a Temp SensorCreated by Ladyada
Testing these sensors is pretty easy but you'll need a battery pack or power supply.

Connect a 2.7-5.5V power supply (2-4 AA batteries work fantastic) so that ground is connected to pin 3 (right pin), and power is connected to pin 1 (left pin)

Then connect your multimeter in DC voltage mode to ground and the remaining pin 2 (middle). If you've got a TMP36 and its about room temperature (25°C), the voltage should be about 0.75V. Note that if you're using a LM35, the voltage will be 0.25V



The sensor is indicating that the temperature is 26.3°C also known as 79.3°F
You can change the voltage range by pressing the plastic case of the sensor with your fingers, you will see the temperature/voltage rise.



With my fingers on the sensor, heating it up a little, the temperature reading is now 29.7°C / 85.5°F
Or you can touch the sensor with an ice cube, perferrably in a plastic bag so it doesn't get water on your circuit, and see the temperature/voltage drop.



I pressed an ice-cube against the sensor, to bring the temperature down to 18.6°C / 65.5°F
Using a Temp SensorCreated by Ladyada
Connecting to a Temperature Sensor

These sensors have little chips in them and while they're not that delicate, they do need to be handled properly. Be careful of static electricity when handling them and make sure the power supply is connected up correctly and is between 2.7 and 5.5V DC - so don't try to use a 9V battery!

They come in a "TO-92" package which means the chip is housed in a plastic hemi-cylinder with three legs. The legs can be bent easily to allow the sensor to be plugged into a breadboard. You can also solder to the pins to connect long wires. If you need to waterproof the sensor, you can see below for an Instructable for how to make an excellent case.

Reading the Analog Temperature Data

Unlike the FSR or photocell sensors we have looked at, the TMP36 and friends doesn't act like a resistor. Because of that, there is really only one way to read the temperature value from the sensor, and that is plugging the output pin directly into an Analog (ADC) input.



Remember that you can use anywhere between 2.7V and 5.5V as the power supply. For this example I'm showing it with a 5V supply but note that you can use this with a 3.3v supply just as easily. No matter what supply you use, the analog voltage reading will range from about 0V (ground) to about 1.75V.

If you're using a 5V Arduino, and connecting the sensor directly into an Analog pin, you can use these formulas to turn the 10-bit analog reading into a temperature:

Voltage at pin in milliVolts = (reading from ADC) * (5000/1024) 
This formula converts the number 0-1023 from the ADC into 0-5000mV (= 5V)

If you're using a 3.3V Arduino, you'll want to use this:

Voltage at pin in milliVolts = (reading from ADC) * (3300/1024) 
This formula converts the number 0-1023 from the ADC into 0-3300mV (= 3.3V)

Then, to convert millivolts into temperature, use this formula:

Centigrade temperature = [(analog voltage in mV) - 500] / 10
Simple Thermometer

This example code for Arduino shows a quick way to create a temperature sensor, it simply prints to the serial port what the current temperature is in both Celsius and Fahrenheit.

Copy Code

1. //TMP36 Pin Variables
2. int sensorPin = 0; //the analog pin the TMP36's Vout (sense) pin is connected to
3.                         //the resolution is 10 mV / degree centigrade with a
4.                         //500 mV offset to allow for negative temperatures
5. /*
6.  * setup() - this function runs once when you turn your Arduino on
7.  * We initialize the serial connection with the computer
8.  */
9. void setup()
10. {
11.   Serial.begin(9600);  //Start the serial connection with the computer
12.                        //to view the result open the serial monitor 
13. }
14. void loop()                     // run over and over again
15. {
16.  //getting the voltage reading from the temperature sensor
17.  int reading = analogRead(sensorPin);  
18.  // converting that reading to voltage, for 3.3v arduino use 3.3
19.  float voltage = reading * 5.0;
20.  voltage /= 1024.0; 
21.  // print out the voltage
22.  Serial.print(voltage); Serial.println(" volts");
23.  // now print out the temperature
24.  float temperatureC = (voltage - 0.5) * 100 ;  //converting from 10 mv per degree wit 500 mV offset
25.                                                //to degrees ((voltage - 500mV) times 100)
26.  Serial.print(temperatureC); Serial.println(" degrees C");
27.  // now convert to Fahrenheit
28.  float temperatureF = (temperatureC * 9.0 / 5.0) + 32.0;
29.  Serial.print(temperatureF); Serial.println(" degrees F");
30.  delay(1000);                                     //waiting a second
31. }
Getting Better Precision

For better results, using the 3.3v reference voltage as ARef instead of the 5V will be more precise and less noisy

This example from the light&temp datalogging tutorial has a photocell but you can ignore it

Note we've changed the TMP36 to A1


To use the 3.3v pin as your analog reference, don't forget to specify "analogReference(EXTERNAL)" in your setup as in the code below:

Copy Code

1. /* Sensor test sketch
2.   for more information see http://www.ladyada.net/make/logshield/lighttemp.html
3.   */
4. #define aref_voltage 3.3         // we tie 3.3V to ARef and measure it with a multimeter!
5. //TMP36 Pin Variables
6. int tempPin = 1;        //the analog pin the TMP36's Vout (sense) pin is connected to
7.                         //the resolution is 10 mV / degree centigrade with a
8.                         //500 mV offset to allow for negative temperatures
9. int tempReading;        // the analog reading from the sensor
10. void setup(void) {
11.   // We'll send debugging information via the Serial monitor
12.   Serial.begin(9600);   
13.   // If you want to set the aref to something other than 5v
14.   analogReference(EXTERNAL);
15. }
16. void loop(void) {
17.   tempReading = analogRead(tempPin);  
18.   Serial.print("Temp reading = ");
19.   Serial.print(tempReading);     // the raw analog reading
20.   // converting that reading to voltage, which is based off the reference voltage
21.   float voltage = tempReading * aref_voltage;
22.   voltage /= 1024.0; 
23.   // print out the voltage
24.   Serial.print(" - ");
25.   Serial.print(voltage); Serial.println(" volts");
26.   // now print out the temperature
27.   float temperatureC = (voltage - 0.5) * 100 ;  //converting from 10 mv per degree wit 500 mV offset
28.                                                //to degrees ((volatge - 500mV) times 100)
29.   Serial.print(temperatureC); Serial.println(" degrees C");
30.   // now convert to Fahrenheight
31.   float temperatureF = (temperatureC * 9.0 / 5.0) + 32.0;
32.   Serial.print(temperatureF); Serial.println(" degrees F");
33.   delay(1000);
34. }
Problem reading two sensors with analogRead()

by Len17 » Sat Jun 20, 2009 4:11 pm

I've been having trouble using analogRead() to read values from two separate sensors. If I run a sketch that calls analogRead() for either one of the sensor inputs I get good results, but if my sketch has two analogRead() calls for both sensors, the values returned are inconsistent and incorrect.

Is there a problem with calling analogRead() for two different input pins in the same sketch?

Details:

I connected two sensors to separate analog input pins. The sensors are a CdS photocell and an LM35 temperature sensor. They're hooked up as described in http://www.ladyada.net/learn/sensors/.

I wrote a simple sketch to read the sensors and print the results out the serial port. The two analogRead() calls seem to interfere with each other. The temperature value is obviously incorrect, and varies wildly as I change the light level sensed by the photocell. But if I simply remove the analogRead() call for the photocell, the temp sensor value is perfectly steady and correct.

Here's the test sketch:

void setup()
{
  Serial.begin(57600);
}

void loop()
{
  int nLight = 0;
  long nTemp = 0;
  nLight = analogRead(0); // remove this line to make the temp sensor work right!
  nTemp = analogRead(5) * 5000L / 1024L  / 10;
  Serial.print("Light = ");
  Serial.print(nLight);
  Serial.print("  Temp = ");
  Serial.println(nTemp);
  delay(250);
}
Here's the sketch's output. You can see the temp sensor change when I obscure the light sensor.


CODE: SELECT ALL
Light = 598  Temp = 20
Light = 600  Temp = 20
Light = 601  Temp = 20
Light = 599  Temp = 20
Light = 532  Temp = 22
Light = 302  Temp = 27
Light = 265  Temp = 27
Light = 231  Temp = 27
Light = 223  Temp = 27


Here's the same test with the first analogRead() removed (so no light value). Now the temp value is constant and correct. (Yes, it's pretty warm in here today.)


CODE: SELECT ALL
Light = 0  Temp = 27
Light = 0  Temp = 27
Light = 0  Temp = 27
Light = 0  Temp = 27
Light = 0  Temp = 27
Light = 0  Temp = 27
Light = 0  Temp = 27
Light = 0  Temp = 27
Light = 0  Temp = 27
La solución……

The problem is that the Atmega on the Arduino has one ADC that is multiplexed for all the analog pins.
When you do an analogRead(), a multiplexer connects the pin you are reading to the ADC. This works fine for low impedance voltage sources.

It takes time for a high impedance sensor like your temperature sensor to change the voltage at the ADC after this switch of pins. Temperature sensors must use low power and thus be high impedance to avoid IR heating.

Try the following:



CODE: SELECT ALL
  analogRead(5);
  delay(10);
  nTemp = analogRead(5) * 5000L / 1024L  / 10;


The first analogRead(5) will switch the pin to the ADC. The delay will allow the voltage at the ADC to stabilize and the second analogRead(5) should get a stable value.
Esta es otra refeencia usando un microarduino…
Experimental schematic
The VCC pin of LM35 connects to 5V,GND connect to GND,and VOUT connect to A0.







Experiment One
· Use Microduino-Ft232RL power

· Program

void setup() {
   Serial.begin(115200);         //Set the baud rate 115200
   pinMode(A0, INPUT); 

}
void loop() {
  int n = analogRead(A0);    //Read the voltage from A0 port
  double vol = n * (5.0 / 1024.0*100);   //Use double type variable to save the temperature value which converted from voltage
  Serial.println(vol);                   //Output the temperature from serial
    delay(1000);                       //Wait for 1s to refresh data
}
The data in serial monitor：






Experiment two
· Use Microduino internal reference source 1.1V to calcuate temperature value

· Program

void setup() {
  Serial.begin(115200);         //Set the baud rate 115200
  analogReference(INTERNAL);  //Invoke 1.1V onboard power souce
}
void loop() {
  int n = analogRead(A0);    //Read the voltage from A0 port
  double vol = n * (1.1 / 1024.0*100);    //Use double type variable to save the temperature value which converted from voltage
  Serial.println(vol);                   //Output the temperature from serial
  delay(1000);                        //Wait for 1s to refresh data
}
The data in serial monitor：






Experiment three
· Use Microduino internal reference source 3.3V to calcuate temperature value

Note：Use the aref pin as the base source, connect the 3.3V base source to aref pin of Microduino






Use the 3.3V as the external base source, please refer to：http://www.geek-workshop.com/thread-5717-1-1.html
· Program

void setup() {
  Serial.begin(115200);         //Set the baud rate 115200
 analogReference(EXTERNAL);  //Use the aref pin as the base source, connect the 3.3V base source to aref pin
}
void loop() {
  int n = analogRead(A0);    //Read the voltage from A0 port
  double vol = n * (3.3 / 1024.0*100);   //Use double type variable to save the temperature value which converted from voltage
  Serial.println(vol);                   //Output the temperature from serial
  delay(1000);                           //Wait for 1s to refresh data
}
The data in serial monitor：






Analysis
Use three methods to test the temperature and got three different value.

· Use the USB power, the result is beter. The power voltage is 4.99V, deviating from the 5 v voltage is 0.2%

· Use internal 1.1V and 3.3V souce, after a bulk sample, the average error range of 1.1V internal reference source is about 1.22%, up 2.2%.

· Under 5V range, the Arduino ADC sampling precision is 10bit that is 1024 level, minimum resolution is 5/1024=0.0048828125V=4.8828125mV

· Under 3.3V internal source range, the Arduino ADC sampling precision is 10bit and the minimum resolution is 3.3/1024=0.00322265625V=3.22265625mV

· Under 1.1V internal source range, the Arduino ADC sampling precision is 10bit and the minimum resolution is 1.1/1024=0.00107421875V=1.07421875mV

· For LM35, every 10mv represents 1℃, so under 5V range, resolution close to 0.5 degrees, Under 3.3V range resolution close to 0.3 degrees, Under 1.1V range resolution close to 0.1 degrees.

Result
According to the above analysis, use USB power supply, the reference source error is smaller then other power relatively. USB power supply temperature closer to the real temperature, when the power supply isn't stable, the internal reference source will play a role. If require a higher precision, then need to use the external precision reference source. Referance error is smaller, and the accuracy of the ADC is higher too. According to the requirement to choose suitable reference source in all kinds of production is also very important.

Practica de Lm35 y processing….

Arduino LM35 Sensor 

My son Paul and I recently finished a project using the Arduino Diecimila microcontroller in conjunction with the Processing open source programming environment to monitor temperature.

The project contains 3 parts:
1. The Arduino board with sensor circuit.
2. The Arduino program.
3. The Processing program.

The Arduino Board with Sensor Circuit
The Arduino circuit board is connected to the LM35 Celsius Temperature sensor. Here is a picture of the project circuit with illustrated wires connected to the temperature sensor.



We used the on board power source (5v and Gnd) to power the LM35 and analog pin 0 (zero) to read the analog output from the sensor. Here's a picture of the circuit wired on a breadboard.




The LM35 temperature sensor's pin-out and package information is as follows:



The Arduino Program

The open-source Arduino environment allows us to write code and load it onto the Arduino board's memory. The development environment is written in Java and based on Processing, avr-gcc, and other open source software.

The Arduino code loops every second to read output from the LM35, converting the analog output into Celsius and sending the data to the computer via a serial communication connection (USB).

Here's the code used to run the Arduino board:

//declare variables
float tempC;
int tempPin = 0;

void setup()
{
Serial.begin(9600); //opens serial port, sets data rate to 9600 bps
}

void loop()
{
tempC = analogRead(tempPin);           //read the value from the sensor
tempC = (5.0 * tempC * 100.0)/1024.0;  //convert the analog data to temperature
Serial.print((byte)tempC);             //send the data to the computer
delay(1000);                           //wait one second before sending new data
}

The Processing Program
The software client portion of this project runs on a PC and is written in Processing. Processing is a language and development environment similar to Arduino and designed for creating visual effects programs. We used Processing to create a small client that can read the serial data from the Arduino board and display the temperature on a slider and in both Celsius and Fahrenheit. We also added a rolling 100 data point graph to display historical temperature data. Here's a screen shot of the Processing application:




Here is the code used for the visual portion of the project:

//import Serial communication library
import processing.serial.*;

//init variables
Serial commPort;
float tempC;
float tempF;
int yDist;
PFont font12;
PFont font24;
float[] tempHistory = new float[100];

void setup()
{
  //setup fonts for use throughout the application
  font12 = loadFont("Verdana-12.vlw"); 
  font24 = loadFont("Verdana-24.vlw"); 
  
  //set the size of the window
  size(210, 200);
  
  //init serial communication port
  commPort = new Serial(this, "COM10", 9600);
  
  //fill tempHistory with default temps
  for(int index = 0; index<100; index++)
    tempHistory[index] = 0;
}

void draw()
{
  //get the temp from the serial port
  while (commPort.available() > 0) 
  {
    tempC = commPort.read();
  
    //refresh the background to clear old data
    background(123);

    //draw the temp rectangle
    colorMode(RGB, 160);  //use color mode sized for fading
    stroke (0);
    rect (49,19,22,162);
    //fade red and blue within the rectangle
    for (int colorIndex = 0; colorIndex <= 160; colorIndex++) 
    {
      stroke(160 - colorIndex, 0, colorIndex);
      line(50, colorIndex + 20, 70, colorIndex + 20);
    }
    
    //draw graph
    stroke(0);
    fill(255,255,255);
    rect(90,80,100,100);
    for (int index = 0; index<100; index++)
    {  
      if(index == 99)
        tempHistory[index] = tempC;
      else
        tempHistory[index] = tempHistory[index + 1];
      point(90 + index, 180 - tempHistory[index]); 
    }
  
    //write reference values
    fill(0,0,0);
    textFont(font12); 
    textAlign(RIGHT);
    text("212 F", 45, 25); 
    text("32 F", 45, 187);
  
    //draw triangle pointer
    yDist = int(160 - (160 * (tempC * 0.01)));
    stroke(0);
    triangle(75, yDist + 20, 85, yDist + 15, 85, yDist + 25);
  
    //write the temp in C and F
    fill(0,0,0);
    textFont(font24); 
    textAlign(LEFT);
    text(str(int(tempC)) + " C", 115, 37);
    tempF = ((tempC*9)/5) + 32;
    text(str(int(tempF)) + " F", 115, 65);
  }
}

Para quitar error de fonts
The sketch needs to a contain a second folder called "data".  Put the font files in there.  If you use the Tools/CreateFont menu command, you'll find that's where they go.

